
	Reason for change:
(

	1)  TTCN sets new RAC for cell geran_cell25. Current TTCN implementation does not update system information type 13 in geran_cell25. This implementation will result in UE not performing the RAU procedure step 26 – 27. 

2). P-TMSI information element is expected in RAU request message – step 25. According to spec. 24.008 chap 9.4.14.5: P-TMSI is implemented for UMTS only. This parameter should be set to omit.

3), lLMEId is set wrongly in RAU accept message – step 26. Template cas_G_LLC_UnitData_Req sets parameter lLMEId from geran_cell24 by default (if input parameter is not present). In this case geran_cell25 is used and is needed add input parameter – geran_cell25.
4), TTCN expects RAU complete message with parameter “E-UTRAN inter RAT handover information“ – step 27. According to 24.008 chap. 9.4.16.3 – This IE shall be included if network has required in RAU accept message. TC specification does not require this specific information in RAU complete message. 
5), TTCN expects trigger before postamble function from EUTRA component. This is not correct implementation because TTCN sent trigger to EUTRA in step 27. EUTRA component receives trigger from GERAN and performs cell release procedure - according to implementation. Any other trigger is not needed. 



	
	

	Summary of change:
(

	1) Add new variable v_SI13RO - update system information type 13 with new RAC value.

2) Replace cr_MobileIdTMSI_lv(v_PTMSI1) with omit.
3)  Add one more input parameter “geran_cell25”.
4) Only template cr_G_RA_UpdComplete should be expected by TTCN – step 27.
5) Remove function f_IRAT_WaitForCoOrd_Trigger at the end of TC before postamble function..

Change 1

	Testcase name
	9.2.3.1.6

	Reason for change
	1)  TTCN sets new RAC for cell geran_cell25. Current TTCN implementation does not update system information type 13 in geran_cell25. This implementation will result in UE not performing the RAU procedure step 26 – 27. 

2). P-TMSI information element is expected in RAU request message – step 25. According to spec. 24.008 chap 9.4.14.5: P-TMSI is implemented for UMTS only. This parameter should be set to omit.

3), lLMEId is set wrongly in RAU accept message – step 26. Template cas_G_LLC_UnitData_Req sets parameter lLMEId from geran_cell24 by default (if input parameter is not present). In this case geran_cell25 is used and is needed add input parameter – geran_cell25.
4), TTCN expects RAU complete message with parameter “E-UTRAN inter RAT handover information“ – step 27. According to 24.008 chap. 9.4.16.3 – This IE shall be included if network has required in RAU accept message. TC specification does not require this specific information in RAU complete message. 
5), TTCN expects trigger before postamble function from EUTRA component. This is not correct implementation because TTCN sent trigger to EUTRA in step 27. EUTRA component receives trigger from GERAN and performs cell release procedure - according to implementation. Any other trigger is not needed. 



	Summary of change
	1) Add new variable v_SI13RO - update system information type 13 with new RAC value.

2) Replace cr_MobileIdTMSI_lv(v_PTMSI1) with omit.
3)  Add one more input parameter “geran_cell25”.
4) Only template cr_G_RA_UpdComplete should be expected by TTCN – step 27.
5) Remove function f_IRAT_WaitForCoOrd_Trigger at the end of TC before postamble function.

	Source of change
	NAS_TrackingArea_UG_GERAN.ttcn


Before change:

	  function f_TC_9_2_3_1_6_GERAN() runs on GERAN_PTC

  {

    var B3_Type v_UpdateType := '000'B;

    var G_LLC_DATAMESSAGE_IND v_LLCDataInd;

    var octetstring v_PTMSI1 := px_PTMSI_Def;

    var octetstring v_PTMSI2 := v_PTMSI1 or4b 'D5555555'O; // first 2 bits must be 1, don't care about the rest

    var octetstring v_LAC1, v_RAC1, v_RAC2;

    var CoOrd_GUTIParam_Type v_GUTI;      

    f_GERAN_Init(EUTRA_GERAN); // This is still only a 2 PTC test

    v_LAC1 := f_GERAN_LAC_Get (geran_Cell24);

    v_RAC1 := f_GERAN_RAC_Get (geran_Cell24);

    v_RAC2 := f_GERAN_RAC_Get (geran_Cell25);

    v_RAC2 := int2oct (oct2int(v_RAC2) +1, 1); // The RAC on cell7 must be different, but by default they're the same

    f_GERAN_RAC_Set (geran_Cell25, v_RAC2);

    f_GERAN_CreateCell (geran_Cell24);

    f_GERAN_CreateCell (geran_Cell25);

    f_GERAN_InterRAT_InitialiseAuthParams (); // @sic R5s110791 sic@

    f_IRAT_WaitForCoOrd_Trigger (EUTRA);

    f_GERAN_TestBody_Set( true );

    //@siclog "Steps 1 - 6" siclog@

    // This is not really the place to call Preamble, but this function does everything we need

    // - set cell to Serving cell, switch on UE, Attach, Authentication, update EUTRA with new params

    f_GERAN_Preamble (geran_Cell24);

    //@siclog "Step 7 - 8" siclog@

    f_GERAN_ActivatePDPContext( geran_Cell24 );

    // in case the UE sends an XID once the bearer is established

    f_GERAN_LLC_XID (geran_Cell24, false);

    //@siclog "Step 9" siclog@

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_NonSuitableCell);

    v_GUTI.MME_GroupId := oct2bit(v_LAC1);

    v_GUTI.MME_Code := substr (oct2bit(v_PTMSI1), 8, 8); // @sic R5s110791 sic@

    v_GUTI.M_TMSI := '11'B & substr (oct2bit(v_PTMSI1), 2, 6) &  oct2bit(v_RAC1) & substr (oct2bit(v_PTMSI1), 16, 16); // @sic R5s110791 sic@

    f_GERAN_SendAuthParameters (EUTRA, v_GUTI);//Send latest Auth params to EUTRA component

    //@siclog "Step 13" siclog@

    // Set Cell 5 to "Serving Cell"

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_ServingCell);

    //@siclog "Step 14" siclog@

    // This function name refers to Attach, but as it only checks for Ch Req, we can use it here too

    f_GERAN_CheckNoAttachOnCell (geran_Cell24, 90.0, "Step 14"); // @sic R5-115769 sic@

    //@siclog "Step 15" siclog@

    // Check: Does the UE respond to paging on Cell 24 with P-TMSI-1 for PS domain?

    f_GERAN_PagewithPTMSI (geran_Cell24, v_PTMSI1);

    //@siclog "Step 15A" siclog@

    f_GPRS_ReceiveUplinkMsg (geran_Cell24, ?);  //NOTE 2// 24.008 cl. 4.7.9.1.1 states that UE will send any LLC frame
    f_GERAN_PreliminaryPass (__FILE__, __LINE__, "Step 15A");

    //@siclog "Step 16" siclog@

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_NonSuitableCell);

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);

    //@siclog "Step 18A" siclog@

    // Set Cell 5 to "Serving Cell"

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_ServingCell);

    //@siclog "Step 18B" siclog@

    // This function name refers to Attach, but as it only checks for Ch Req, we can use it here too

    f_GERAN_CheckNoAttachOnCell (geran_Cell24, 90.0, "Step 18B");  // @sic R5s110791 sic@

    //@siclog "Step 19" siclog@

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_NonSuitableCell);

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);

    //@siclog "Step 24" siclog@

    // Set Cell 25 to "Serving Cell"

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    f_GERAN_SetCellPowerLevel (geran_Cell25, tsc_GERAN_ServingCell);

    //@siclog "Step 25" siclog@

    // Check: Does the UE transmit a ROUTING AREA UPDATE REQUEST message with P-TMSI and RAI on cell 25?

    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (geran_Cell25, car_G_LLC_UnitData_IND ( cr_G_RAUReq_Ids ( cr_GMM_UpdateType,

                                                                                                      cr_RAI (omit, cr_LAI_Any_v, v_RAC1),

                                                                                                      cr_PTMSI_SignatureAny ifpresent,

                                                                                                      ?,

                                                                                                      cr_MobileIdTMSI_lv(v_PTMSI1), //NOTE 2
                                                                                                      cr_MobileIdAny('1A'O) ifpresent,

                                                                                                      cr_RoutingAreaId2 ifpresent)));

    v_UpdateType := v_LLCDataInd.unit.msg.rauReq.updateType.updateValue;

    if (v_UpdateType == '000'B) { // @sic R5-115769 sic@

      v_UpdateType := '100'B;

    } else if (v_UpdateType == '001'B) {

      v_UpdateType := '101'B;

    }

    //@siclog "Step 26" siclog@

    // SS sends a ROUTING AREA UPDATE ACCEPT message with P-TMSI2 and RAI2

    f_GPRS_SendDownlinkMsg(geran_Cell25, 1, cas_G_LLC_UnitData_Req (f_GERAN_TLLI_Get(geran_Cell25),

                                                                    tsc_LLCSapi_GMM,

                                                                    tsc_LLC_PM,

                                                                    px_GSM_CipheringOnOff,

                                                                    cs_G_RA_UpdAcc3 (cs_GMM_UpdateResult( v_UpdateType ),

                                                                                     cs_RAI (omit,

                                                                                             cs_LAI(omit, f_GERAN_PLMN_Get(geran_Cell25), f_GERAN_LAC_Get(geran_Cell25)),

                                                                                             v_RAC2 ),

                                                                                     cs_PTMSI_Signature (px_PTMSI_SigDef),

                                                                                     cs_MobileIdTMSIorPTMSI ('18'O, v_PTMSI2)))); //NOTE 3
    //Receive RAU Complete Message

    //@siclog "Step 27" siclog@

    // the UE sends a ROUTING AREA UPDATE COMPLETE message

    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (geran_Cell25, car_G_LLC_UnitData_IND (cr_G_RA_UpdComplete(cr_EUTRAN_HO_InfoAny));  //NOTE 4    

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);

    f_GERAN_TestBody_Set( false );

    // Test case finished, so take cell down

    f_IRAT_WaitForCoOrd_Trigger (EUTRA);  //NOTE 5
    f_GERAN_Postamble ( geran_Cell25, G1_IDLE ); // @sic R5-115769 sic@

    f_GERAN_SS_CellRelease ( geran_Cell24 );

  } // end f_TC_9_2_3_1_6_GERAN


After change :

	  function f_TC_9_2_3_1_6_GERAN() runs on GERAN_PTC

  {

    var B3_Type v_UpdateType := '000'B;

    var G_LLC_DATAMESSAGE_IND v_LLCDataInd;

    var octetstring v_PTMSI1 := px_PTMSI_Def;

    var octetstring v_PTMSI2 := v_PTMSI1 or4b 'D5555555'O; // first 2 bits must be 1, don't care about the rest

    var octetstring v_LAC1, v_RAC1, v_RAC2;

    var CoOrd_GUTIParam_Type v_GUTI;

    var SI13RO v_SI13RO; //Anritsu TTCN CR required //NOTE 1 
    f_GERAN_Init(EUTRA_GERAN); // This is still only a 2 PTC test

    v_LAC1 := f_GERAN_LAC_Get (geran_Cell24);

    v_RAC1 := f_GERAN_RAC_Get (geran_Cell24);

    v_RAC2 := f_GERAN_RAC_Get (geran_Cell25);

    v_RAC2 := int2oct (oct2int(v_RAC2) +1, 1); // The RAC on cell7 must be different, but by default they're the same

    f_GERAN_RAC_Set (geran_Cell25, v_RAC2);

    v_SI13RO := f_GERAN_SysInfo_GetSI13_RestOctets (geran_Cell25); //Anritsu TTCN CR required //NOTE 1
    v_SI13RO.rAC := v_RAC2; //Anritsu TTCN CR required //NOTE 1
    f_GERAN_SysInfo_SetSI13_RestOctets (geran_Cell25, v_SI13RO); //Anritsu TTCN CR required   //NOTE 1
    f_GERAN_CreateCell (geran_Cell24);

    f_GERAN_CreateCell (geran_Cell25);

    f_GERAN_InterRAT_InitialiseAuthParams (); // @sic R5s110791 sic@

    f_IRAT_WaitForCoOrd_Trigger (EUTRA);

    f_GERAN_TestBody_Set( true );

    //@siclog "Steps 1 - 6" siclog@

    // This is not really the place to call Preamble, but this function does everything we need

    // - set cell to Serving cell, switch on UE, Attach, Authentication, update EUTRA with new params

    f_GERAN_Preamble (geran_Cell24);

    //@siclog "Step 7 - 8" siclog@

    f_GERAN_ActivatePDPContext( geran_Cell24 );

    // in case the UE sends an XID once the bearer is established

    f_GERAN_LLC_XID (geran_Cell24, false);

    //@siclog "Step 9" siclog@

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_NonSuitableCell);

    v_GUTI.MME_GroupId := oct2bit(v_LAC1);

    v_GUTI.MME_Code := substr (oct2bit(v_PTMSI1), 8, 8); // @sic R5s110791 sic@

    v_GUTI.M_TMSI := '11'B & substr (oct2bit(v_PTMSI1), 2, 6) &  oct2bit(v_RAC1) & substr (oct2bit(v_PTMSI1), 16, 16); // @sic R5s110791 sic@

    f_GERAN_SendAuthParameters (EUTRA, v_GUTI);//Send latest Auth params to EUTRA component

    //@siclog "Step 13" siclog@

    // Set Cell 5 to "Serving Cell"

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_ServingCell);

    //@siclog "Step 14" siclog@

    // This function name refers to Attach, but as it only checks for Ch Req, we can use it here too

    f_GERAN_CheckNoAttachOnCell (geran_Cell24, 90.0, "Step 14"); // @sic R5-115769 sic@

    //@siclog "Step 15" siclog@

    // Check: Does the UE respond to paging on Cell 24 with P-TMSI-1 for PS domain?

    f_GERAN_PagewithPTMSI (geran_Cell24, v_PTMSI1);

    //@siclog "Step 15A" siclog@

    f_GPRS_ReceiveUplinkMsg (geran_Cell24, ?);  //NOTE 2// 24.008 cl. 4.7.9.1.1 states that UE will send any LLC frame    

    f_GERAN_PreliminaryPass (__FILE__, __LINE__, "Step 15A");

    //@siclog "Step 16" siclog@

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_NonSuitableCell);

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);

    //@siclog "Step 18A" siclog@

    // Set Cell 5 to "Serving Cell"

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_ServingCell);

    //@siclog "Step 18B" siclog@

    // This function name refers to Attach, but as it only checks for Ch Req, we can use it here too

    f_GERAN_CheckNoAttachOnCell (geran_Cell24, 90.0, "Step 18B");  // @sic R5s110791 sic@

    //@siclog "Step 19" siclog@

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_NonSuitableCell);

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);

    //@siclog "Step 24" siclog@

    // Set Cell 25 to "Serving Cell"

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    f_GERAN_SetCellPowerLevel (geran_Cell25, tsc_GERAN_ServingCell);

    //@siclog "Step 25" siclog@

    // Check: Does the UE transmit a ROUTING AREA UPDATE REQUEST message with P-TMSI and RAI on cell 25?

    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (geran_Cell25, car_G_LLC_UnitData_IND ( cr_G_RAUReq_Ids ( cr_GMM_UpdateType,

                                                                                                      cr_RAI (omit, cr_LAI_Any_v, v_RAC1),

                                                                                                      cr_PTMSI_SignatureAny ifpresent,

                                                                                                      ?,

                                                                                                      omit, //Anritsu TTCN CR required //NOTE 2
                                                                                                      cr_MobileIdAny('1A'O) ifpresent,

                                                                                                      cr_RoutingAreaId2 ifpresent)));

    v_UpdateType := v_LLCDataInd.unit.msg.rauReq.updateType.updateValue;

    if (v_UpdateType == '000'B) { // @sic R5-115769 sic@

      v_UpdateType := '100'B;

    } else if (v_UpdateType == '001'B) {

      v_UpdateType := '101'B;

    }

    //@siclog "Step 26" siclog@

    // SS sends a ROUTING AREA UPDATE ACCEPT message with P-TMSI2 and RAI2

    f_GPRS_SendDownlinkMsg(geran_Cell25, 1, cas_G_LLC_UnitData_Req (f_GERAN_TLLI_Get(geran_Cell25),

                                                                    tsc_LLCSapi_GMM,

                                                                    tsc_LLC_PM,

                                                                    px_GSM_CipheringOnOff,

                                                                    cs_G_RA_UpdAcc3 (cs_GMM_UpdateResult( v_UpdateType ),

                                                                                     cs_RAI (omit,

                                                                                             cs_LAI(omit, f_GERAN_PLMN_Get(geran_Cell25), f_GERAN_LAC_Get(geran_Cell25)),

                                                                                             v_RAC2 ),

                                                                                     cs_PTMSI_Signature (px_PTMSI_SigDef),

                                                                                     cs_MobileIdTMSIorPTMSI ('18'O, v_PTMSI2)),

                                                                    geran_Cell25)); //Anritsu TTCN CR required //NOTE 3
    //Receive RAU Complete Message

    //@siclog "Step 27" siclog@

    // the UE sends a ROUTING AREA UPDATE COMPLETE message

    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (geran_Cell25, car_G_LLC_UnitData_IND(cr_G_RA_UpdComplete)); //Anritsu TTCN CR required //NOTE 4
    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);

    f_GERAN_TestBody_Set( false );

    // Test case finished, so take cell down

    //f_IRAT_WaitForCoOrd_Trigger (EUTRA); //Anritsu TTCN CR required //NOTE 5
    f_GERAN_Postamble ( geran_Cell25, G1_IDLE ); // @sic R5-115769 sic@

    f_GERAN_SS_CellRelease ( geran_Cell24 );

  } // end f_TC_9_2_3_1_6_GERAN

